# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>

pkgname=python-magnumclient
pkgver=4.1.0
_commit=de11f40d0c632ff839e5838ed650d1821c87d8a8
pkgrel=3
pkgdesc="Client library for Magnum API"
arch=('any')
url="https://docs.openstack.org/python-magnumclient/latest/"
license=('Apache')
depends=('python-pbr' 'python-keystoneauth1' 'python-stevedore' 'python-requests'
         'python-oslo-i18n' 'python-oslo-log' 'python-oslo-serialization' 'python-oslo-utils'
         'python-os-client-config' 'python-osc-lib' 'python-prettytable' 'python-cryptography'
         'python-debtcollector')
makedepends=('git')
checkdepends=('python-openstackclient' 'python-oslotest' 'python-osprofiler')
source=("git+https://github.com/openstack/python-magnumclient.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd python-magnumclient
  python setup.py build
}

check() {
  cd python-magnumclient
  stestr run
}

package() {
  cd python-magnumclient
  python setup.py install --root="$pkgdir" --optimize=1
}
